



Hardware:

- Einfacher Druckeranschluss an Kanal A der SIO nur als Ausgabekanal (es können keine Daten vom Drucker empfangen werden)
- Pegelwandler TTL <-> RS232 nötig
 - SIO-A/TxD => Wandler => Drucker/RxD
 - SIO-A/CTS <= Wandler <= Drucker/DTR
 - Masseverbindung, ggf. auch 5P vom LC-80ex
- Drucker mit serieller Schnittstelle auf 2400 Baud eingestellt (Testbeispiel mit K6304: DIL7=on, DIL6=off)
- Es wird die gleiche Konfiguration von CTC und SIO wie beim Terminal benutzt, nur eben Kanal A. Dazu: **Brücke im Adapter zwischen RxTxCB und TxCA der SIO**
- Wenn eine vom Terminal unabhängige Programmierung nötig ist, muss CTC-Kanal 1 benutzt werden. Die Steckbrücke ist dann von JP12/"1" an SIO-A/TxC legen.
- Die bereits beim Laden/Sichern gezeigte herkömmlichen Programmierung der SIO ist für den Drucker lediglich um eine "Bremse" erweitert: Der Drucker meldet mit seiner DTR-Leitung per high-Pegel an CTS des LC-80ex, dass er beschäftigt ist. Dem LC-80ex wird damit die Sendeerlaubnis entzogen. Er hält weitere Daten solange zurück, bis Drucker wieder "bereit" meldet. Das passiert jedoch erst bei größeren Druckmengen, die die Puffergröße im Drucker übersteigen (oder wenn der Drucker nicht eingeschaltet ist).
- JP9: "1/2" = Takt: 1,8 MHz
- kein Interrupt (LC-80ex muss immer auf Drucker warten)

Verdrahtungsschema:



Testaufbau:



Das Funktionsmuster des Druckadapters, fliegend verdrahtet...

Der Adapter wird bei Bedarf auf die Stiftleisten links und rechts der SIO aufgesteckt.

Der Anschluss zum Druckers erfolgt bei mir per DIN-Buchse. Die Belegung entspricht der von Modul 003 am KC85.

Der Test erfolgte an einem Thermodrucker K6304 sowie Nadeldrucker K6319.

Software:

- Die nachfolgend dargestellten Grundroutinen INITDR und DRUCK beinhalten die Initialisierung der SIO und die Ausgabe eines Druckzeichens.
- Neben den Steuerzeichen #0D und #0A werden nur "druckbare" ASCII-Zeichen #20...#7E ausgegeben, alle übrigen Codes werden ignoriert bzw. zu Leerzeichen gewandelt.
- Es gibt in der Testroutine kein "timeout" o.ä.! Wenn der Drucker nicht bereit ist (nicht eingeschaltet, offline, Papier alle, Error,...) hängt der LC-80ex wegen /CTS=high solange in einer angeschobenen Ausgabe fest. Das kann bei Bedarf aber ergänzt werden (siehe unten).

```
;PROGRAMMIERDATEN CTC+SIO:
PERI: DEFB #00000111B ;STEUERWORT CTC: DI/ZEITGEBER/ZK-FOLGT/VT16/RESET
      DEFB #03          ;ZEITKONSTANTE CTC
      DEFB #04          ;ZUGRIFF SIO WR4
      DEFB 01000100B    ;SIO-TEILER=16, 1 STOPBIT, KEINE PARITÄT
      DEFB #03          ;ZUGRIFF SIO WR3
      DEFB 11100000B    ;8 EMPF-BITS, DCD/CTS ALS EMPFANGS-/SENDESTEUERUNG
      DEFB #05          ;ZUGRIFF SIO WR5
      DEFB 01101010B    ;8 SENDEBITS, SENDEFREIGABE, RTS=0
;CTC ZK=3 + SIO T=16 ERGEBEN BEI 1,8 MHZ 2400 BAUD.
;DAS UNTERSTRICHENE BIT BEWIRKT DIE "HARDWARE-BREMSE"
```

```

;INITIALISIERUNG DRUCKER
INITDR:
    LD        HL,PERI        ;AB TABELLENANFANG
    LD        B,03H         ;3 ZEICHEN
    LD        C,ST CTC0     ;AN CTC0 (#EC), HIER GGF. CTC1 BENUTZEN!
    DI
    OTIR                      ;SENDEN
    LD        B,06H         ;FOLGENDE 6 ZEICHEN
    LD        C,ST_SIOA     ;AN SIO (#DE STEUERW. KANAL A)
    OTIR                      ;SENDEN
    EI
    ;NUN CRLF AN DRUCKER AUSGEBEN, KANN HIER AUCH ENTFALLEN
    LD        A,0DH         ;
    CALL      DRUCK         ;AUSGEBEN
    LD        A,0AH         ;JE NACHDEM, WIE DRUCKER SELBST EINGESTELLT IST,
    CALL      DRUCK         ;KEIN ZUSÄTZLICHES 0A NÖTIG...
    RET

```


```

;AUSGABE ZEICHEN IN A AN DRUCKER
DRUCK: PUSH    AF
    PUSH      DE
    CP        #0D           ;CR?
    JR        Z,DRU1        ;JA
    CP        #0A           ;LF?
    JR        Z,DRU1        ;JA
    ;WEITERE BENÖTIGTE STEUERZEICHEN HIER GGF. NOCH BEHANDELN
    ;...
    CP        #20           ;< #20? (SONSTIGES STEUERZEICHEN)
    JR        C,DRU3        ;JA=>IGNORIEREN
    CP        #7E           ;>= #7E?
    JR        NC,DRU4       ;JA=>ERSATZZEICHEN
    ;WAR DRUCKBARES ZEICHEN
DRU1: LD        D,A         ;AUSGABEZEICHEN SICHERN
DRU2: LD        C,ST_SIOA   ;AN SIO (#DE STEUERW. KANAL A)
    IN        A,(C)         ;LESEN RR0
    BIT       2,A           ;SENDEPUFFER LEER?
    JR        Z,DRU2        ;NEIN => WEITER ABFRAGEN!
    ;DIE SCHEIFE WIRD BEI "DRUCKER BESCHÄFTIGT" (/CTS=HIGH) NICHT VERLASSEN,
    ;DA DER PUFFER WEGEN SENDEVERBOTS NICHT LEER WERDEN KANN.
    ;ABHILFE FÜR "TIMEOUT": VORHER RR0/BIT 5 (ZUSTAND CTS) ABFRAGEN UND
    ;ENTSPRECHEND VERZWEIGEN...
    LD        A,D           ;AUSGABEZEICHEN HOLEN
    LD        C,DA_SIOA     ;SIO (#DC DATEN KANAL A)
    OUT       (C),A         ;SENDEN!
    POP       DE
DRU3: POP       AF
    RET
DRU4: LD        A,#20       ;ERSATZZEICHEN FÜR NICHTDRUCKBARES
    JR        DRU1

```

Anwendung:

- Das kleine Testprogramm "dddd.z80" kann per RS232 oder USB geladen werden (hat eine Autostart-Funktion). Erneuter Start mit "ADR 2000 EX" möglich. Es programmiert CTC0 + SIO Kanal A und gibt einen Text an den Drucker aus:



- Die o.a. Grundroutinen für Druckerinitialisierung und Druckausgabe sind in den LCTOOLSV2 in ähnlicher Form integriert. Adressen des Sprungverteilers:
A01E: DRUCKER_INIT ;Initialisieren SIO-A + CTC
A024: AUSGABE ;serielle Ausgabe
A027: DRUCKER_AUS ;Druckausgabe aus \ Setzen/Rücksetzen
A02A: DRUCKER_EIN ;Druckausgabe ein / eines Flags
Ist das Flag gesetzt, so erfolgt bei serieller Zeichenausgabe auf den Bildschirm auch eine zusätzliche Druckausgabe.
- Das "rdk Tiny Basic" als Bestandteil der LCTOOLS-V2 wurde um LLIST, LPRINT und LFILES erweitert.
- Auch das neue "8k-BASIC-Spezial" unterstützt einen Drucker mit den o.a. Anweisungen.

Es wurde hiermit eine prinzipielle Möglichkeit des Druckens demonstriert. Sie erhebt keinen Anspruch auf Komfort. Für den **Lernvorgang** am Lerncomputer LC-80ex sollte das aber ausreichen. Er wurde vor allem Wert auf Nachvollziehbarkeit und Übersichtlichkeit gelegt. Änderungen und Verbesserungen sind selbstverständlich möglich.